Graphics with Processing

2010-04 色彩とピクセル処理

http://vilab.org

塩澤秀和

4.1 色彩

色のデータ形式

- 色の指定方法
  - 1つの数値（グレースケール）
  - 3つの数値の組（カラー）
    初期モードは RGB 各0～255
  - 16進数カラーコード #rrggbb
  - color型の変数
- color型
  - 色を表すデータ型（実態はint）
  - color関数で合成できる
    color(成分1, 成分2, 成分3)
  - 例）color c = color(r, g, b);
- 成分の取得
  - red(c), green(c), blue(c), hue(c), saturation(c), brightness(c), alpha(c)

半透明の表現

- アルファ値（p.225）
  - 色の第4成分（透過処理用）
  - 重ね塗りでの色の混合率
  - 例）c = color(r, g, b, a);
  - 例）fill(255, 0, 0, 128);

色モードの設定

- colorMode(モード, 値範囲)
  - モード：カラーモデル
    RGB または HSB
  - 値範囲：成分の上限値
    - colorMode(モード, 範囲1, 範囲2, 範囲3) の形式もある
  - 例）colorMode(HSB, 1.0);
  - サンプル Basics → Color

4.2 表色系/カラーモデル(p.201)

RGBカラーモデル

- 光の三原色(赤, 緑, 青)





HSB(HSV/HSI)カラーモデル

- 光の三属性
  - 色相(H):色あい
  - 彩度(S):あざやかさ
  - 明度(B/V/I):明るさ
  - メニュー Tools → Color Selector



図はWikipediaから

4.3 ピクセル処理

ピクセル配列による操作

- ピクセルとは（p.11）
  - 画面を構成する画素1点1点（pixel ← picture cell）

  ⇒ ラスター表現のグラフィックス
- pixels[]
  - 各画素の色（color型のデータ）を格納する1次元配列
  - 画面座標(x, y)の要素は pixels[y * width + x]
- loadPixels()
  - ピクセル処理の開始処理
  - 画面の画素ごとの色データをpixels[]に読み込む
- updatePixels()
  - pixcels[]を画面に書き戻す

ピクセル配列の操作

- ピクセルの読み出し
  - color c;
  - c = pixels[y * width + x];
- ピクセルの書き込み
  - pixels[y * width + x] = c;

ピクセル配列を使わない操作

- copy(x1, y1, w1, h1, x2, y2, w2, h2)
- copy(画像, $x_{画像}$, $y_{画像}$, $w_{画像}$, $h_{画像}$, x, y, w, h)
  - 領域や画像からのコピー
- get(), get(x, y, 幅, 高さ)
  - 表示内容を画像として取得

4.4 ピクセル配列


pixels[0]
pixels[1]
pixels[x]
pixels[width-1]
pixels[width]
x
loadPixels()
pixels[y*width]
(x, y)
updatePixels()
y
画面から
ピクセル
配列へ
pixels[y*width+x]
ピクセル
配列から
画面へ
pixels[(height-1)*width]
pixels[height*width-1]

4.5 ラスター化

- ラスター化（p.207）
  - ベクター表現（座標とパラメータ）の図形を，画素の集合に変換



直線のラスター化の例

  - このサンプルをさらに高速化
    ⇒ ブレゼンハムのアルゴリズム

```
void draw() {
  background(0);
  loadPixels();
  if (mouseX > mouseY)
    pxline(0, 0, mouseX, mouseY);
  updatePixels();
}
```

```
void pxline(int x1, int y1,
            int x2, int y2)
{
  color c = color(255, 255, 255);

  float d = (float)(y2-y1)/(x2-x1);
  float e = 0.0;

  int x = x1, y = y1;
  while (x <= x2) {

    pixels[y * width + x] = c;

    x++;
    e += d;
    if (e >= 0.5) {
      e -= 1.0;
      y++;
    }
  }
}
```

xが1増えるあたりのyの増分（小数）

本来のy座標との累積誤差

画素設定

xが1増えるごとにyの誤差はd増加

yの誤差が0.5以上になったらyを1増やす

4.6 画像データ

画像データ

- 画像の利用
  - サンプル Basics → Image → LoadDisplayImage など
  - 対応形式: jpg gif png tga
- PImage型
  - 画像を扱うには，PImage型のグローバル変数を用意しておく

    PImage img;
- loadImage("ファイル名")
  - 画像データの読み込み
  - 通常，setup()で1回だけ行う

    img = loadImage("a.jpg")
  - ファイルは，事前にメニューのSketch → Add File...でデータフォルダにコピーしておく

画像表示

- image(画像, x, y)
- image(画像, x, y, 幅，高さ)
  - 画像の描画（拡大・縮小）
- imageMode(モード)
  - rectMode/elipseModeと同様

図形データ

- ベクター（ベクトル）表現
  - 座標とパラメータによる図形表現
  - 対応形式: SVG
- PShape, loadShape, shape, shapeMode ⇒ ほぼ画像と同様
  - サンプル Basics → Shape → LoadDisplayShape など

4.7 オブジェクト指向基礎

オブジェクト指向

- オブジェクトとは
  - データとその操作をセットにして，使いやすくしたもの
  - 例）PImage img

オブジェクト指向用語

- 「クラス」:オブジェクトの型
  - 例）PImage
- 「インスタンス」:オブジェクト変数
  - 例）img
- 「フィールド」:オブジェクトの属性
  - 例）img.height
- 「メソッド」:オブジェクトの操作
  - 例）img.save()

PImage型の例

- フィールド
  - img.width, img.height
    - 画像のサイズ（横・縦の幅）
  - img.pixels[]
    - 画像データのピクセル配列
- メソッド（一部）
  - img.save("ファイル名")
    - 画像にファイル名をつけて保存
  - img.get(x, y, 幅, 高さ)
    - 画像の一部を画像として取り出す
  - img.resize(幅, 高さ)
    - 画像のサイズを変更する
  - img.loadPixels(), img.updatePixels()
    - ピクセル処理のためのメソッド

4.8 演習課題

準備課題

- 右のプログラムに適切な setup 関数を補って，実行してみなさい
  - マウスをドラッグしてみなさい
  - さらに，★の範囲を
    if (random(1.0) < 0.2)
    というif文で囲んでみなさい

提出課題

- 上から下に流れる模様を，下から上に流れるように変更しなさい
  - 下→上以外のものは認めない
  - できたら，円のかわりに画像を表示させてみるとよい
- ヒント
  - (x,y+1)を(x,y)にコピーする
  - yのfor文も変える必要がある

```
void draw() {
  color c;
  loadPixels();
  for (int x = 0; x < width; x++) {
    // ★
    for (int y = height-1; y > 0; y--) {
      // (x,y-1)の色を(x,y)にコピー
      c = pixels[(y - 1) * width + x];
      pixels[y * width + x] = c;
    }
    // 最上行には新しい色を入れる
    pixels[0 * width + x] =
        color(0, 0, frameCount % 256);
    // ★
  }
  updatePixels();

  if (mousePressed) {
    noStroke();
    fill(255, 220, 220, 200);
    ellipse(mouseX, mouseY, 20, 20);
  }
}
```